

Pfaffian 形式への変換

```

import("yang.rr")$
extern Lo$
yang.define_ring([x,y]);
L1o=dx-dy;
L2o=dx^2-(x+y);
Lo=[L1o,L2o];

def pf() {
  L=Lo;
  L=yang.util_pd_to_euler(L,[x,y]);
  L=map(nm,L);
  L=map(dp_ptod,L,[dx,dy]);
  G=yang.buchberger(L);
  print(yang.stdmon(G));
  S1=yang.constant(1);
  Sy=yang.operator(y);
  Base=[S1,Sy];
  Pf=yang.pfaffian(Base,G);
  return Pf;
/* F_x = Pf[0]*F, F_y = Pf[1]*F */
}

def bynd() {
  L=Lo;
  dp_gr_print(1);
  M = newmat(6,4,
    [[0,0,1,1],
     [0,0,0,-1],
     [0,0,-1,0],
     [1,1,0,0],
     [0,-1,0,0],
     [-1,0,0,0]]);

  G=nd_weyl_gr(L,[x,y,dx,dy],0,M);
  return G;
}

print("pf(); bynd(); ")$
end$

```

テキストエディタで入力を書いて, asir の中で load("ファイル名"); と入力してプログラムを読み込む. Windows のテキストエディタはアクセサリ/メモ帳 (notepad) など. Unix のテキストエディタは emacs など.

```

/* cf. misc-2009/10/fish/bess.rr */
import("yang.rr")$
extern Lo$
yang.define_ring([x,y]);
L1o=dx*dy+1;
L2o=dx^2-2*x*dx+2*y*dy+2*a;
L3o=2*y*dy^2+2*(a+1)*dy-dx+2*x;
Lo=[L1o,L2o,L3o];

def besspf() {
  L=base_replace(Lo,[[a,1/2]]);
  L=yang.util_pd_to_euler(L,[x,y]);
  L=map(nm,L);
  L=map(dp_ptod,L,[dx,dy]);
  G=yang.buchberger(L);
  yang.stdmon(G);
  S1=yang.constant(1);
  Sx=yang.operator(x);
  Sy=yang.operator(y);
  Base=[S1,Sx,Sy];
  Pf=yang.pfaffian(Base,G);
  return Pf;
/* F_x = Pf[0]*F, F_y = Pf[1]*F */
}

def bynd() {
  L=base_replace(Lo,[[a,1/2]]);
  dp_gr_print(1);
  M = newmat(6,4,
    [[0,0,1,1],
     [0,0,0,-1],
     [0,0,-1,0],
     [1,1,0,0],
     [0,-1,0,0],
     [-1,0,0,0]]);

  G=nd_weyl_gr(L,[x,y,dx,dy],0,M);
  return G;
}

print("besspf(); bynd(); ")$
end$

/*
besspf();
[ [ 0 (1)/(x) 0 ]
  [-x (2*x^2+1)/(x) -2*x ]
  [-y 0 0 ] [ 0 0 (1)/(y) ]
  [-x 0 0 ]
  [-x (1/2)/(x) (-1/2)/(y) ] ]

bynd();
3.4.nd_gb done.
...
[-dx+2*y*dy^2+3*dy+2*x, dy*dx+1, dx^2-2*x*dx+2*y*dy+1]
*/

```